﻿/*

  The contents of this file are subject to the Mozilla Public License Version
  1.1 (the "License"); you may not use this file except in compliance with
  the License. You may obtain a copy of the License at 
  
           http://www.mozilla.org/MPL/ 
  
  Software distributed under the License is distributed on an "AS IS" basis,
  WITHOUT WARRANTY OF ANY KIND, either express or implied. See the License
  for the specific language governing rights and limitations under the License. 
  
  The Original Code is AST'r Framework.
  
  The Initial Developer of the Original Code is
  ALCARAZ Marc (aka eKameleon) <ekameleon@gmail.com>.
  Portions created by the Initial Developer are Copyright (C) 2004-2011
  the Initial Developer. All Rights Reserved.
  
  Contributor(s) :
  
*/
package egallery.advanced.net.remoting.gallery
{
    import egallery.advanced.logging.logger;
    import egallery.advanced.models.PictureModel;
    import egallery.advanced.vo.PictureVO;

    import system.remoting.RemotingService;
    import system.remoting.RemotingServiceListener;
    
    /**
     * This remoting service listener is invoked when the server-side gallery service is launched. 
     */
    public class GetGallery extends RemotingServiceListener
    {
        /**
         * Creates a new GetGallery instance.
         * @param service The RemotingService reference of this listener.
         */
        public function GetGallery( service:RemotingService=null ) 
        {
            super( service ) ;
            PictureVO.register() ;
        }
        
        /**
         * The picture model reference.
         */
        public var model:PictureModel ;
        
        /**
         * Invoked if the Remoting connection is success.
         */
        public override function result( result:* , service:RemotingService ):void 
        {
            if ( model.size() > 0 )
            {
                model.clear() ;
            } 
            logger.info( this + " result " + result ) ;
            if ( result != null && result is Array )
            {
                var size:int = result.length ;
                logger.info( this + " run " + size + " PictureVO objects" ) ;
                if ( size > 0 )
                {
                    for (var i:int ; i< size ; i++)
                    {
                        model.addVO( result[i] as PictureVO ) ;
                    }
                    if( model.size() > 0 )
                    {
                        model.run() ;
                    }
                }
                else
                {
                    logger.warn( this + " no picture found.") ;
                }
            }
            else
            {
                logger.warn( this + " run failed with an empty result.") ;
            }
        }
    }
}